
//***************************************************************
//* MACRO: UTLNGPRV.WCM
//* PURPOSE:  Provide a function that allows the user to switch "Gamma
//*                              Productions"  UniType program  language to the
//*                              "Previous UniType Language" without leaving WordPerfect.
//* In addition to switching the UniType language the
//* font at the cursor insertion point is switched to the 
//* ILI configured UI font for the new language.
//*
//* This macro is idealy suited for using UniType to create
//* documents where switching between 2 languages is
//* frequent.
//***************************************************************
//* WordPerfect's standard behavior is such that when you change fonts at
//* a cursor insertion point, you are actually changing the font from the cursor
//* point all the way to the next font change in the text.  To override this
//* behavior, this macro records the existing font at the cursor point before
//* making the font change.  This font is stored in the variable: oldFont.  After
//* making the font change to the new font a dummy character and a font  
//* change back to the oldFont are inserted.  Then  the cursor is moved back
//* before the dummy character and the dummy character is deleted.
//*
//* This KLUDGE effective allows the macro to change the font of only
//* new characters that will be typed after evoking the macro.
//*
//* If you want the standard WordPerfect behavior you would edit the 
//* second code line of this macro to say: WordPerfectKludge:=False
//* 
//* Note: If the macro is assigned to a keystroke, pressing that keystroke
//* will NOT run the macro when UniType is active and the UniType
//* language is in "Word" or "Phrase" mode and there is text pending
//* in the UniType edit buffer.
//***************************************************************
//* Application (A1; "WordPerfect"; Default; "US") 
Display(State: Off!)
WordPerfectKludge:=True
DDEInitiate(hChannel; "UniType"; "UniType")
If(hChannel != 0)
 DDEExecute(hChannel; "ut_set_language")
 langStr = DDERequest(hChannel; "ut_get_language")
 uifontStr = DDERequest(hChannel; "ut_get_utfont " + langStr)
 SelectOff()
 CodesRevealed:= ?RevealCodesActive
 If(CodesRevealed = True)
   RevealCodes(Off!)
 EndIf



 oldfontStr = ?Font      
 Font(uifontStr)
 //////////////////////////////////////////////
 // see comment at top of Macro
 If(WordPerfectKludge)
   Type("?")                    
   Font(oldfontStr)             
   PosCharPrevious()      
   DeleteCharNext()        
 EndIf
 //////////////////////////////////////////////
 If(CodesRevealed = True)
   RevealCodes(On!)
 EndIf
 DDETERMINATE(hChannel)
Else
 MessageBox(mbResult; "Warning!"; "Unable to establish DDE connection with UniType." +   
NTOC(0F90Ah) + NTOC(0F90Ah) + "Please run the UniType program and try 
again."; IconExclamation!)
Endif


